Two Level Production System (smod_t2)

1

Outline

- Agents in the models
- *smod_t* model
  - Price tree
  - Production (review)
- Two-Level Production System
  - Price and Quantity trees
  - Top Level CES function
  - Top Level Leontief function
  - Second Level CES function
- Calibration
  - Top level
  - Second level
- Implications

2

cgemod

smod_t

3

cgemod

Agents in smod_t/t2

- 4 commodities
- 4 activities
- 4 factors
- 2 households
- Government – taxes and spends
- Savings/investment – no time dimension
- Trade
  - with a single partner, ROW
  - small country assumption
  - CET and CES for each commodity

Same as for *smod_t*

4

cgemod

Commodity Price System



5

Price Normalisation & Tax Instruments

ALL prices derived relative to BASIC PRICES

$$PE_c = PD_c = PM_c = 1 \qquad \forall c \qquad \text{Basic Prices}$$

$$PQS_c = PXC_c = PX_a = 1 \qquad \forall c = a$$

Instrument	Name	Base Price	Behaviour
Import duties	tm_c	PWM_c	*ad valorem*
Export taxes	te_c	PWE_c	*ad valorem*
Sales taxes (GST)	ts_c	PQS_c	*ad valorem*
Production taxes	tx_a	PX_a	*ad valorem*
Income taxes	tyh_h	YH_h	*ad valorem*

6

cgemod

Production Price System



$WFA_{f,a} = WF_f * WDIST_{f,a}$

7

cgemod

Production Block Equations

$$QX_a = ad_a \prod_f \left(FD_{f,a}\right)^{\alpha_{f,a}}$$ **Production function**

$$FD_{f,a} = \frac{QX_a * PVA_a * \alpha_{f,a}}{WF_f}$$ **1st Order Condition**

$$QINTD_c = \sum_a ioqintdqx_{c,a} * QX_a$$

$ioqintdqx_{c,a}$ = quantity of c per unit of a

Intermediate Input Demand

8

cgemod

cgemod

Two-Level Production Equations

9

cgemod

Production System: Quantities



Additions

1. $QINT_a$
2. σ_1
3. σ_{22}

10

cgemod

cgemod

Production System: Prices



Additions

1. $PINT_a$
2. σ_1
2. σ_{22}

$$WFA_{f,a} = WF_f * WDIST_{f,a}$$

11

cgemod

Top Level CES Equations: 1

$$PX_a = \sum_c ioqxacqx_{a,c} * PXC_c$$

***ioqxacqx*$_{a,c}$**
one to one mapping from a to c

Solves for *PVA*$_a$

$$PX_a * (1 - TX_a) * QX_a = (PVA_a * QVA_a) + (PINT_a * QINT_a)$$

$$PINT_a = \sum_c (ioqtdqd_{c,a} * PQD_c)$$

***ioqtdqd*$_{c,a}$**
quantity of c used by a to produce a unit of $QINT_a$

12

cgemod

Top Level CES Equations: 1 in GAMS

```
PXDEF(a)..
      PX(a)  =E=  SUM(c,ioqxcqx(a,c)*PXC(c)) ;


PVADEF(a)..
   PX(a)*(1-TX(a))*QX(a)
          =E=  PVA(a)*QVA(a))+(PINT(a)*QINT(a)) ;


PINTDEF(a)..
    PINT(a)  =E=  SUM(c,ioqtdqd(c,a)  *  PQD(c)) ;
```


13

Top Level CES Equations: 2

$$AD_a^x = \left[\left(adxb_a + dabadx_a\right) * ADXADJ\right] + \left(DADX * adx01_a\right)$$

$$QX_a = AD_a^x \left(\delta_a^x QVA_a^{-rhoc_a^x} + \left(1-\delta_a^x\right) QINT_a^{-rhoc_a^x}\right)^{-\frac{1}{rhoc_a^x}} \qquad \forall aqx_a$$

$$\frac{QVA_a}{QINT_a} = \left[\frac{PINT_a}{PVA_a} * \frac{\delta_a^x}{\left(1-\delta_a^x\right)}\right]^{\frac{1}{\left(1+rhoc_a^x\right)}} \qquad \forall aqx_a$$

14

Top Level CES Equations: 2 in GAMS

```
ADXEQ(a)..
     ADX(a) =E= ((adxb(a)+dabadx(a))*ADXADJ)
                   +(DADX*adx01(a)) ;
```

ADX(a)
Efficiency variable

```
QXPRODFN(a)$aqx(a)..
     QX(a) =E= ADX(a)*(deltax(a)*QVA(a)**(-rhocx(a))
                  + (1-deltax(a))*QINT(a)**(-rhocx(a)))
                     **(-1/rhocx(a)) ;


QXFOC(a)$aqx(a)..
     QVA(a) =E= QINT(a)*((PINT(a)/PVA(a))
                  *(deltax(a)/(1-deltax(a))))
                     **(1/(1+rhocx(a))) ;
```

aqx(a) subset of a with σ at level 1

15

Top Level Leontief Equations

$QINT_a = ioqintqx_a * QX_a \quad \forall aqxn_a$

```
QINTDEF(a)$aqxn(a)..
  QINT(a) =E= ioqintqx(a) * QX(a) ;
```

$QVA_a = ioqvaqx_a * QX_a \quad \forall aqxn_a$

```
QVADEF(a)$aqxn(a)..
  QVA(a) =E= ioqvaqx(a) * QX(a) ;
```

aqxn(a) complement to aqx(a)

$QINTD_c = \sum_a ioqtdqd_{c,a} * QINT_a$

```
* Intermediate Input Demand

QINTDEQ(c)..
  QINTD(c) =E= SUM(a,ioqtdqd(c,a)
                  *QINT(a)) ;
```

$QXC_c = \sum_a ioqxac_{a,c} * QX_a$

```
* Commodity Output

COMOUT(c)..
  QXC(c) =E= SUM(a,ioqxcqx(a,c)
                  *QX(a)) ;
```


16

Second-Level Equations

$$AD_a^{va} = \left[\left(advab_a + dabadva_a\right) * ADVAADJ\right] + \left(DADVA * adva01_a\right)$$

$$QVA_a = AD_a^{va} * \left[\sum_{f\$\delta_{f,a}^{va}} \delta_{f,a}^{va} * AD_{f,a}^{fd} * FD_{f,a}^{-\rho_a^{va}}\right]^{-1/\rho_a^{va}}$$

$$WF_f * WFDIST_{f,a} = PVA_a * AD_a^{va} * \left[\sum_{f\$\delta_{f,a}^{va}} \delta_{f,a}^{va} * AD_{f,a}^{fd} * FD_{f,a}^{-\rho_a^{va}}\right]^{-\left(\frac{1+\rho_a^{va}}{\rho_a^{va}}\right)} * \delta_{f,a}^{va} * FD_{f,a}^{\left(-\rho_a^{va}-1\right)}$$

$$= PVA_a * QVA_a * AD_a^{va} * \left[\sum_{f\$\delta_{f,a}^{va}} \delta_{f,a}^{va} * AD_{f,a}^{fd} * FD_{f,a}^{-\rho_a^{va}}\right]^{-1}$$

$$* \delta_{f,a}^{va} * \left(AD_{f,a}^{fd}\right)^{-\rho_a^{va}} * \delta_{f,a}^{va} * FD_{f,a}^{\left(-\rho_a^{va}-1\right)}$$

Note

1. $WF_f * WFDIST_{f,a}$ on *LHS*
2. *n*-argument form
3. Substitution for QVA_a

17

Second-Level Equations - GAMS

```
ADVAEQ(a)..
 ADVA(a) =E= ((advab(a)+dabadva(a))*ADVAADJ)
                +(DADVA*adva01(a)) ;

QVAPRODFN(a)..
 QVA(a) =E= ADVA(a)*(SUM{f$deltava(f,a),deltava(f,a)
              *[ADFD(f,a)*FD(f,a)]**[-rhocva(a)]})
                   **(-1/rhocva(a)) ;

QVAFOC(f,a)$deltava(f,a)..
 WF(f)*WFDIST(f,a)
   =E= PVA(a)*QVA(a)*(SUM{fp$deltava(fp,a),deltava(fp,a)
             *[ADFD(f,a)*FD(fp,a)]**[-rhocva(a)]})**(-1)
                *deltava(f,a)*ADFD(f,a)**(-rhocva(a))
                   *FD(f,a)**(-rhocva(a)-1) ;
```

ADFD(f,a)
Stock-flow variable

$deltava(f,a) Only implement if deltava(f,a) >0

18

cgemod

Calibration

19

cgemod

Calibration: Initial values for variables

```
QX0(a)                      = SAM("total",a)/PX0(a) ;
QXC0(c)                     = SUM(a,SAM(a,c))/PXC0(c) ;
```

PX0(a) = PXC0(c) = 1.0

```
QINT0(a)$SUM(c,SAM(c,a))= SUM(c,SAM(c,a)/PQD0(c)) ;
```

PQD0(c) – derived price from PQS0(c) = 1.0

```
QVA0(a)                     = SUM(f,SAM(f,a)) ;

PINT0(a)$(QINT0(a)) = SUM(c,(SAM(c,a)/PQD0(c)/QINT0(a))*PQD0(c));

PVA0(a)$QVA0(a)             = SUM(f,SAM(f,a))/QVA0(a) ;

QINTD0(c)$SUM(ap,SAM(c,ap)) = SUM(a,SAM(c,a)/PQD0(c)) ;
```

Value Quantities
QX0(a); QXC0(c); QINT0(a); QVA0(a); QINTD(c)

All values based on
1. Transaction values
2. Price normalisation

20

cgemod

Calibration: Initial values for variables

```
FD0(f,a)            = FACTUSE(f,a) ;

FS0(f)              = SUM(a,FACTUSE(f,a)) ;

WF0(f)$(FS0(f))     = SUM(a,SAM(f,a))/FS0(f) ;

WFDIST0(f,a)$FD0(f,a)
                    =(SAM(f,a)/FD0(f,a))/WF0(f) ;

WFDIST0(f,a)$(FD0(f,a) EQ 0)          = 0.0 ;
```

FACTUSE(f,a)
'Real' quantities or VALUE quantities

WF0(f) and WFDIST0(f,a)
Value quantities = 1.0
'Real' quantities = ?? (scaling)

21

Calibration: Level 1 CES

```
rhocx(a)    = (1/ELASTX(a,"sigmax")) - 1 ;

predeltax(a)$(QINT0(a))
  = (PVA0(a)/PINT0(a))*(QVA0(a)/QINT0(a))**(1+rhocx(a)) ;

deltax(a)   = predeltax(a)/(1.0+predeltax(a)) ;

ADX0(a)$deltax(a)
 = QX0(a)/(deltax(a)*QVA0(a)**(-rhocx(a))
   +(1-deltax(a))*QINT0(a)**(-rhocx(a)))**(-1/rhocx(a)) ;
```

ELASTX(a,"sigmax") - **exogenous**

Other values only use TVs

```
ADXADJ0         = 1 ;
DADX0           = 0 ;
adxb(a)         = ADX0(a) ;
dabadx(a)       = 0.0 ;
adx01(a)        = 0.0 ;
```

ppf **vv observed 'efficiency'**

22

Calibration: Level 1 Leontief

```
use(c,a)$SAM("TOTAL",a)        = SAM(c,a)/SAM("TOTAL",a) ;

comactco(c,a)$(QX0(a)$PQD0(c))
                    = (SAM(c,a)/PQD0(c))/QX0(a) ;

ioqintqx(a)$QX0(a)     = QINT0(a)/QX0(a) ;

ioqvaqx(a)$QX0(a)      = QVA0(a)/QX0(a) ;

* Intermediate Input Demand

ioqtdqd(c,a)$(QINT0(a)$PQD0(c))
                      = SAM(c,a)/PQD0(c)/QINT0(a) ;

* Activity Output
ioqxcqx(a,c)$acx(a)    = SAM(a,c)/SUM(cp,SAM(a,cp))
```

Calibration uses only TVs and derived prices

23

Calibration: Level 2 CES

```
rhocva(a)       = (1/ELASTX(a,"sigmava")) - 1 ;
```

ELASTX(a,"sigmava") - **exogenous**

```
deltava(f,a)$SAM(f,a)
       = (WFDIST0(f,a)*WF0(f)*(FD0(f,a))**(1+rhocva(a)))
            /SUM(fp,WFDIST0(fp,a)*WF0(fp)*(FD0(fp,a))
                    **(1+rhocva(a))) ;
```

Other values only use TVs

```
ADVA0(a)
  = QVA0(a)/(SUM(f$(FD0(f,a)),deltava(f,a)*FD0(f,a)
            **(-rhocva(a))))**(-1/rhocva(a));

ADVAADJ0        = 1 ;
DADVA0          = 0 ;
advab(a)        = ADVA0(a) ;
dabadva(a)      = 0.0 ;
adva01(a)       = 0.0 ;
ADFD0(f,a)      = 1 ;
```

***ppf* vv observed 'efficiency'**

ADFD(f,a) Stock-flow variable

24

cgemod

Factor Market Clearing

25

cgemod

Factor Market Clearing

Relative to *smod_t*

```
* Technology for CES production functions Level 1 of nest

 ADXADJ.FX          = ADXADJ0 ;
 DADX.FX            = DADX0 ;

* Technology for CES aggregation functions Level 2 of nest

 ADVAADJ.FX         = ADVAADJ0 ;
 DADVA.FX           = DADVA0 ;

* Technology for factor & activity specific efficiency

 ADFD.FX(f,a)       = ADFD0(f,a) ;
```

26

cgemod

cgemod

Implications

27

cgemod

Implications

- PVA_a changes relative to $PINT_a$
  - Induces changes in QVA_a and $QINT_a$
- PQD_c changes for one or more intermediate
  - Induces changes in $PINT_a$
- $ADVA_a$ changes
  - Induces changes in PVA_a
- $ADFD_{f,a}$ changes for one or more (f,a)
  - Induces changes in
    - WF_f*$WFDIST_{f,a}$
    - PVA_a

Leontief implies sigma = 0.0

ELASTX(a,"sigmax"/"sigmava") and deltaxa/deltavaa determine responsiveness

28

cgemod

cgemod

The End

Two Level Production System (*smod_t2*)

29

cgemod